
Not-so-simple SQL Server Stored Procedure 
 
 
CREATE Procedure SetFillDate 
   @ordernum varchar(8) 
As 
 
/* procedure SetFillDate  
   by Alan G. Labouseur 
      3NF Consulting, Inc. 
*/ 
 
begin tran 
 
declare @allItems int 
declare @nullItems int 
declare @clearFilled bit 
declare @setFilled bit 
declare @gotAll int 
declare @currentFillDate datetime 
 
set @clearFilled = 0 
set @setFilled = 0 
 
-- Get the number of line items for this order. 
select @allItems = count(*) 
   from orderdet 
   where order_num = @ordernum 
 
-- Get the number if "quantity is null" line items. 
select @nullItems = count(*) 
   from Coed..orderdet 
   where order_num = @ordernum and  
      quantity is null 
 
-- Check to see if they are all null 
if @nullItems = @allItems begin 
   -- All items are currently null 
   set @clearFilled = 1 
end else begin  
   -- At least some items have been filled.  
   -- See if everything that been ordered has been filled. 
   select @gotAll = count(*)  
   from Coed..Orderdet 
   where order_num = @ordernum and  
     quantity > 0 and  
     filled < quantity 
   if @gotAll = 0 begin 
      set @setFilled = 1 
   end else begin 
      set @clearFilled = 1 
   end -- if 
end -- if 
 
if @setFilled = 1 begin 



   -- We have all the line items we've asked for. 
   -- Check to see whether there is already a fill date in there. 
   select @currentFillDate = fill_date 
   from Coed..Ordermas 
   where order_num = @ordernum 
   if @currentFillDate is null begin 
      set @setFilled = 1 
   end else begin 
      set @setFilled = 0 
   end -- if 
end -- if 
 
if @setFilled = 1 begin 
   -- If we're here, we rilly rilly want to set the fill date. 
   update Coed..Ordermas 
   set fill_date = GETDATE() 
   where order_num = @ordernum 
end else begin 
   if @clearFilled = 1 begin 
      update Coed..Ordermas 
      set fill_date = null 
      where order_num = @ordernum 
   end -- if @clearFilled 
end -- if 
 
commit tran 


